

微型计算机

BASIC语言 常用程序库



北京工业大学

内 容 简 介

本书是普及计算方法和语言程序的工具书。把常用的数值计算方法编成 BASIC 语言标序，内容包括插值、拟合与平滑，数值微分，数值积分，超越方程求解，特殊函数，常方程，线性代数方程求解，矩阵计算，特征值与特征向量计算，一维最优化方法等。程序在国产 MDR-Z80 型微处理机上通过。

为方便读者，每个子程序前面都有该子程序所用数学方法简述及使用说明，其后均有例题程序。

本书可供拥有 MDR-Z80 微型处理机（或 PS-80、TRS-80 等微型机）的工程技术人员及有关专业人员作为手册查阅，也可作为理工科大专院校师生的参考资料，对于 BASIC 的初学者也是有益的参考书。

前 言

近年来, 计算机技术取得了惊人的发展。微型机的应用以其成本低、体积小和功能不断完善而迅速渗透和充实了各行各业的技术领域。其作用与成就正日益卓著。

利用微型机可对产品进行优化设计; 存贮测量信息; 对测量结果进行分析、综合和作各种处理, 并在系统工程信息管理工作中得到了广泛的应用。因而在教学中以及在各个技术领域的科技人员中普及和提高微型计算机的应用水平, 更好地发挥微型机的效益, 已成为当务之急。其中除学习一些微型计算机的基本知识外, 首先是要掌握一种以至几种计算机算法语言, 编制应用程序。

在几种常用的计算机算法语言中, BASIC 语言是比较简单易学的, 而且 BASIC 语言已配置在各种微型机上。在用计算机算法语言进行程序编制时, 既要了解数值计算方法, 又要熟悉算法语言, 需要花费很多劳动。本书将一些常用计算方法编成标准程序, 并力求简练、易懂、准确, 以利学生在学习算法语言及作课程设计时作为参考, 并期望能帮助各行各业参考本书的人, 能很快掌握使用计算机的本领, 将使电子计算机这个二十世纪中出现的最先进的设备和技术在祖国四化建设中发挥更大的效益。

各种微型机所用 BASIC 语言略有差异, 如有的有自定义函数, 有的就没有等, 需查阅有关机型的使用说明书。本书根据中国科学院数学研究所主编的“BASIC 语言常用算法程序汇编”改编, 以适应国产 MDR-Z80 型微处理机, 以及 TRS-80、PS-80、PS-85 等微型机用户的需要。对原书中一种方法有多个程序的部分, 原则上只改编一个程序; 改正了原书的印刷错误, 并结合 MDR-Z80 型微处理机对某些语句作了改动; 因该机无矩阵语句, 故在第七章加入一节正方矩阵求逆子程序; 第九章改成一维最优化方法, 因该机软件已有快速富氏变换, 故在此不再编入。每个子程序前面都有该子程序所用数学方法简述和该子程序使用说明, 后面均有例题, 对如何调用该子程序作了具体示范。使用者只要给出具体数据, 按照例题的式样写出主程序, 即可上机算题。

本书的附录对本书所用的 BASIC 语言作了简单介绍, 建议在阅读本书时先看一下, 以便更好地参考本书各子程序。

参加本书编写工作的同志有张汉萍、白钰鹏、姜凯生、赵洪德。

由于我们水平有限, 且时间仓促, 书中一定存在不少缺点和错误, 敬请读者批评指正。

编者 1983.12

目 录

第一章 插值、拟合及平滑

第一节	牛顿向前、向后插值	1
第二节	拉格朗日插值	6
第三节	埃特金插值	8
第四节	一元三点插值	10
第五节	二元三点插值	12
第六节	爱尔米特插值	15
第七节	三次自然样条函数插值、微商及积分	16
第八节	最小二乘曲线拟合	22
第九节	参数表达的曲线用周期样条函数拟合	25
第十节	五点三次平滑	30
第十一节	最小二乘多项式逼近	33

第二章 数值积分

第一节	定步长辛普生求积	38
第二节	变步长辛普生求积	40
第三节	龙贝格求积	42
第四节	高斯法求多重积分	46

第三章 超越方程

第一节	下降法解非线性方程组	50
第二节	对分区间套法解超越方程	53
第三节	牛顿法求高次代数方程全部实根	55
第四节	拟牛顿法解非线性方程组	57

第四章 特殊函数

第一节	二项式系数	63
第二节	阶乘	64
第三节	整数阶贝塞尔函数	65
第四节	Γ 函数	66
第五节	Γ 函数的对数	68
第六节	用多项式逼近误差函数	70
第七节	第一类完全椭圆积分	72

第八节	第二类完全椭圆积分	73
第五章 常微分方程		
第一节	定步长龙格——库塔方法	74
第二节	变步长定点输出龙格——库塔方法	77
第三节	折线方法和改进的折线方法	82
第四节	定步长哈明方法	85
第五节	变步长定点输出哈明方法	91
第六节	变步长定点输出单步方法	100
第七节	病态常微分方程组的数值积分	106
第六章 线性代数方程组求解		
第一节	主元素消去方法	112
第二节	高斯塞德尔迭代方法	114
第三节	追赶方法	117
第四节	系微矩阵对称正定时适用的改进平方根方法	120
第五节	病态方程组的迭代方法	125
第七章 矩阵计算		
第一节	求矩阵元素的极大值与极小值	130
第二节	求行列式的值	132
第三节	求任意正方矩阵的逆矩阵	134
第四节	求两矩阵的乘积矩阵	136
第五节	求任意矩阵的转置矩阵	138
第八章 特征值和特征向量计算		
第一节	求对称矩阵特征值和特征向量的雅可比方法	140
第二节	实对称三对角矩阵的 QL 方法	144
第三节	化对称带型矩阵为三对角阵的吉文方法	152
第四节	用豪斯霍尔德变换化对称矩阵为对称三对角阵	158
第五节	用二分法计算实对称三对角阵的特征值	161
第九章 一线最优化方法		
第一节	0.618 法求极值	167
第二节	二次插值法求极值	169
附 录		173

第一章 插值、拟合及平滑

第一节 牛顿向前、向后插值

一、数学方法简述

1. 给定函数 $y=f(x)$ 在等距结点 $x_k=x_0+kh$ 上的函数值为 $y_k=f(x_k), k=0,1,2,\dots, N$, 利用牛顿向前或向后插值公式算出被插值点 x 上的函数值。

2. 利用差分公式算出 n 阶差分

① 向前差分公式

$$\Delta^n y_k = \Delta^{n-1} y_{k+1} - \Delta^{n-1} y_k$$

其中 $n=1,2,\dots, k=0,1,\dots, N-1$ 。

② 向后差分公式

$$\nabla^n y_k = \nabla^{n-1} y_k - \nabla^{n-1} y_{k-1}$$

其中 $n=1,2,\dots, k=N, N-1,\dots, 1$ 。

3. 牛顿向前插值公式

令 $t=(x-x_0)/h$, 即 $x=x_0+th$, 则有

$$\begin{aligned} f(x) &= f(x_0+th) = y_0 + \frac{t}{1!} \Delta y_0 + \frac{t(t-1)}{2!} \Delta^2 y_0 + \dots \\ &\quad + \frac{t(t-1)(t-2)\dots(t-n+1)}{n!} \Delta^n y_0 \end{aligned}$$

4. 牛顿向后插值公式

令: $x=x_N+th$, 则有

$$\begin{aligned} f(x) &= f(x_N+th) = y_N + \frac{t}{1!} \nabla y_N + \frac{t(t+1)}{2!} \nabla^2 y_N + \dots \\ &\quad + \frac{t(t+1)(t+2)\dots(t+n-1)}{n!} \nabla^n y_N \\ &= y_N + \frac{t}{1!} \Delta y_{N-1} + \frac{t(t+1)}{2!} \Delta^2 y_{N-2} + \dots \\ &\quad + \frac{t(t+1)(t+2)\dots(t+n-1)}{n!} \Delta^n y_{N-n} \end{aligned}$$

二、子程序使用说明

根据差分的特性可知,当差分阶数达到某个值时,例如 $r+1$ 阶差分,则有: $|\Delta^{r+1}y_k| < 2^r \times 10^{-m}$, 其中 m 为 y_k 所具有的小数位数;而所有小于等于 r 阶的差分都不满足上述条件。向后差分也有类似的特性。我们在使用差分时,只应使用到 r 阶差分。为此,我们在本节子程序中引用了控制参数 $E \leq 2^r \times 10^{-m}$ 。

当被插值点 x 给定后,本子程序选取最靠近 x 的五个节点进行插值,当所给的 x 在区间 (x_i, x_{i+1}) 上,且 $i+r(=4) \leq N$ 时,则使用牛顿向前插值公式;当 $i+r > N$ 时,则使用牛顿向后插值公式。

1. 简单变量

M ——给定的被插值点的个数减一;

N ——给定的等距结点的个数减一;

H ——给定的等距结点间的长度;

E ——控制差分的阶数,使 $r+1$ 阶差分的多数(8/10)差分值满足 $|\Delta^{r+1}y_k| \leq E$;

P ——差分的阶数(本子程序最高为四阶);

N_2 ——初值为给定的等距结点的个数,为程序中计算某阶差分的差分值个数。

2. 数组

$X(N)$ ——存放给定的结点值;

$Y(N)$ ——存放给定结点上的函数值;

$Z(M)$ ——存放给定的被插值点的值;

$A(N)$ ——存放一阶差分值;

$B(N)$ ——存放二阶差分值;

$C(N)$ ——存放三阶差分值;

$D(N)$ ——存放四阶差分值;

$T(N)$ ——计算差分子程序中,存放各阶差分值;

$Q(N)$ ——计算差分子程序中,存放前一阶差分值;

$F(M)$ ——存放被插值点的函数值;

$R(P)$ ——存放插值多项式的各阶差分值。

3. 数据排列顺序

$M, N, H, E, P, N_2, X(N), Y(N), Z(M)$

4. 计算结果存放在 $F(M)$ 中。

5. 在主程序的数据语句中,按数据排列顺序给出具体数值后,由 GOSUB1000 转本子程序。

三、子程序

1000 'BASIC, NEWTON

1005 READ M, N, H, E, P, N₂

1010 DIM X(N), Y(N), Z(M), A(N), B(N), C(N), D(N), T(N), Q(N),

```

      F(M), R(P)
1015 FOR I=0 TO N:READ X(I):NEXT
1016 FOR I=0 TO N:READ Y(I):NEXT
1017 FOR I=0 TO M:READ Z(I):NEXT
1020 FOR I=0 TO N
1025 Q(I)=Y(I)
1030 NEXT
1035 GOSUB 1235
1040 FOR I=0 TO N
1045 A(I)=T(I):PRINTA(I),
1050 NEXT I
1060 K1=INT(N2*.9)
1065 IF N0>K1 THEN 1280
1070 FOR I=0 TO N
1075 Q(I)=A(I)
1080 NEXT
1085 GOSUB 1235
1090 FOR I=0 TO N
1095 B(I)=T(I):PRINTB(I),
1100 NEXT
1110 K1=INT(N2*.9)
1115 IF N0>K1 THEN 1280
1120 FOR I=0 TO N
1125 Q(I)=B(I)
1130 NEXT
1135 GOSUB 1235
1140 FOR I=0 TO N
1145 C(I)=T(I):PRINTC(I),
1150 NEXT
1160 K1=INT(N2*.9)
1165 IF N0>K1 THEN 1280
1170 FOR I=0 TO N
1175 Q(I)=C(I)
1180 NEXT
1185 GOSUB 1235
1190 FOR I=0 TO N

```



```

1195 D(I)=T(I):PRINTD(I),
1200 NEXT
1210 K1=INT(N2*.9)
1215 IF N0>K1 THEN 1280
1220 PRINT "UJ"
1230 GOTO 1280
1235 N1=N1+1
1240 N2=N2-1
1245 N0=0
1250 FOR I=0 TO N2
1255 T(I)=Q(I+1)-Q(I)
1257 X1=T(I)
1260 IF ABS(T(I))>E THEN 1270
1265 N0=N0+1
1270 NEXT
1275 RETURN
1280 FOR J=0 TO M
1285 FOR I=0 TO N-1
1290 IF X(I)>Z(J) THEN 1325
1295 IF X(I+1) <Z(J) THEN 1315
1300 S=I
1305 R(1)=A(S)
1310 GOTO 1345
1315 NEXT
1325 PRINT "BUZI"
1330 STOP
1335 INPUT Z(J)
1340 GOTO 1285
1345 IF S+N1>N THEN 1370
1350 R(2)=B(S)
1355 R(3)=C(S)
1360 R(4)=D(S)
1365 GOTO 1385
1370 R(2)=B(S-1)
1375 R(3)=C(S-2)
1380 R(4)=D(S-3)
1385 K=1

```

```

1390 H1=1
1395 IF S+N1>N THEN 1415
1400 F(J)=Y(S)
1405 T1=(Z(J)-X(S))/H
1410 GOTO 1425
1415 F(J)=Y(S+1)
1420 T1=(Z(J)-X(S+1))/H
1425 FOR I=1 TO N1
1430 IF S+N1>N THEN 1445
1435 K=K*(T1-I+1)
1440 GOTO 1450
1445 K=K*(T1+I-1)
1450 H1=H1*I
1455 F(J)=F(J)+K/H1*R(I)
1460 NEXT I
1465 NEXT J
1470 FOR I=0 TO M:LPRINTF(I):NEXT
1475 RETURN

```

四、例 题

1. 给定等距结点 $x_k = 4000 + k \times 500$ 及对应的函数值 y_k (y_k 的具体数值见主程序中的对应数据), $k=0, 1, 2, \dots, 14$ 。求被插值点 Z_j (具体数值见主程序) 上的函数值 $f(z_j) = ?$ $j=0, 1, 2$ 。

2. 主程序

```

5 DATA 2, 14, 500, .1, 4, 14
10 DATA 4000, 4500, 5000, 5500, 6000, 6500, 7000, 7500, 8000, 8500, 9000,
    9500, 10000, 10500, 11000
15 DATA 1.38, 1.48, 1.58, 1.69, 1.81, 1.94, 2.1, 2.28, 2.5, 2.76, 3.06,
    3.41, 3.83, 4.33, 4.93
20 DATA 5200, 7200, 10700
25 GOSUB 1000
30 END

```

3. 计算结果

$F(5200) = 1.6228$

$F(7200) = 2.1672$

$F(10700) = 4.558$

第二节 拉格朗日插值

一、数学方法简述

1. 设已知函数 $y(x)$ 的结点值为 $x_i (i=0, 1, \dots, n)$ 及对应的函数值为 $y_i (i=0, 1, \dots, n)$; 对于给定的不是结点的 x , 用一元拉格朗日 $n+1$ 点插值公式计算其对应的函数值 $y(x)$ 。

2. 插值公式如下

$$y(x) = \sum_{j=0}^n \prod_{\substack{i=0 \\ i \neq j}}^n \left(\frac{x-x_i}{x_j-x_i} \right) y_j$$

3. 本节子程序是对 $m+1$ 个不是结点的值 $x_k (k=0, 1, \dots, m)$ 进行成组插值求出其对应的函数值 $y_k = y(x_k)$ 。

二、子程序使用说明

1. 简单变量

N——给定的插值结点个数减一;

M——要插值的函数个数减一;

2.——数组

X(N)——存放给定的插值结点值;

Y(N)——存放给定插值结点上的函数值;

U(M)——存放被插值点的结点值;

V(M)——存放插值结果。

3. 数据排列顺序

N, M, X(N), Y(N), U(M)

4. 计算结果存在 V(M) 中。

5. 在主程序的数据语句中, 按数据排列顺序给出数值后, 由 GOSUB1000 转本节子程序。

三、子程序

```
1000 'LAGRANGE INSERT VALUE
1010 READ N, M
1020 DIM X(N), Y(N), U(M), V(M)
1030 FOR I=0 TO N:READ X(I):NEXT
1040 FOR I=0 TO N:READ Y(I):NEXT
```

```

1050 FOR I=0 TO M:READ U(I):NEXT
1060 FOR K=0 TO M
1070 FOR J=0 TO N
1080 L=1
1090 FOR I=0 TO N
1100 IF I=J THEN 1120
1110 L=L*(U(K)-X(I))/(X(J)-X(I))
1120 NEXT I
1130 V(K)=V(K)+L*Y(J)
1140 NEXT J
1150 LPRINT "V( "K" )=" V(K)
1160 NEXT K
1170 RETURN

```

四、例 题

1. 已知函数如表 1-1 所示

表 1-1

x	-2	-0.4	0.2	1	4
y	24	-0.2688	-0.0768	0	480

2. 求当 $x = -1.5, -1, -0.2, 0, 0.4, 0.8, 1.5, 2$ 时对应的函数值。

3. 主程序

```

5 DATA 4, 7
10 DATA -2, -.4, .2, 1, 4, 24, -.2688, -.0768, 0, 480
15 DATA -1.5, -1, -.2, 0, .4, .8, 1.5, 2
20 GOSUB 1000
25 END

```

4. 计算结果

```

V(0)=5.625
V(1)=0
V(2)=-.0768
V(3)=0
V(4)=-.2688
V(5)=-.4608
V(6)=5.625
V(7)=24

```

第三节 埃特金插值

一、数学方法简述

1. 已知函数 $f(x)$ 的 $n+1$ 个结点 $a_i (i=0, 1, \dots, n)$ 及其对应的 $n+1$ 个函数值 $y_i (i=0, 1, \dots, n)$ 。

设 $I_0(x)=y_0, I_1(x)=y_1, \dots, I_n(x)=y_n$

令

$I_{0,1}(x)$: 为用 a_0, a_1 作插值点的一次插值多项式;

$I_{0,2}(x)$: 为用 a_0, a_2 作插值点的一次插值多项式;

$I_{1,2}(x)$: 为用 a_1, a_2 作插值点的一次插值多项式;

.....

$I_{0,1,2}(x)$: 为用 a_0, a_1, a_2 作插值点的二次插值多项式;

$I_{0,1,3}(x)$: 为用 a_0, a_1, a_3 作插值点的二次插值多项式;

$I_{0,1,\dots,k}(x)$: 为用 $a_0, a_1, \dots, a_k$ 作插值点的 k 次插值多项式。

有下列关系式成立

$$\begin{aligned} I_{0,1}(x) &= \frac{1}{a_1 - a_0} \begin{vmatrix} I_0(x) & x - a_0 \\ I_1(x) & x - a_1 \end{vmatrix} \\ &= \frac{(x - a_0)y_1 - (x - a_1)y_0}{a_1 - a_0} \\ I_{0,1,2}(x) &= \frac{1}{a_1 - a_2} \begin{vmatrix} I_{0,1}(x) & x - a_1 \\ I_{0,2}(x) & x - a_2 \end{vmatrix} \\ &= \frac{1}{a_0 - a_1} \begin{vmatrix} I_{0,2}(x) & x - a_0 \\ I_{1,2}(x) & x - a_1 \end{vmatrix} \\ &\dots\dots \\ I_{0,1,\dots,k}(x) &= \frac{1}{a_q - a_p} \begin{vmatrix} I_{0,1,\dots,p-1,p+1,\dots,k}(x) & x - a_q \\ I_{0,1,\dots,q-1,q+1,\dots,k}(x) & x - a_p \end{vmatrix} \end{aligned}$$

这样, 可从 $I_0, I_1, \dots$ 出发, 逐步求出 $I_{0,1}(x), I_{0,2}(x), I_{1,2}(x), \dots, I_{0,1,\dots,k}(x)$, 成为一系列线性插值问题。

二、子程序使用说明

1. 简单变量

N ——已知插值结点的个数减一;

X ——被插值点的值;

F ——存放 x 点的函数值。

2. 数组

A(N)——存放已知插值结点的值;

Y(N)——存放已知插值结点对应的函数值。

3. 数据排列顺序

N, X, A(N), Y(N)

4. 在主程序的数据语句中,按数据排列顺序给出具体数值后,由 GOSUB 1000 转本节子程序

三、子程序

```
1000 'AITKEN INSERT VALUE
1010 READ N, X
1020 DIM A(N), Y(N)
1030 FOR I=0 TO N:READ A(I):NEXT
1040 FOR I=0 TO N:READ Y(I):NEXT
1050 FOR J=0 TO N-1
1060 FOR I=J+1 TO N
1070 Y(I)=((X-A(J))*Y(I)-(X-A(I))*Y(J))/(A(I)-A(J))
1080 NEXT I, J
1090 F=Y(N)
1100 LPRINT "F(" X ")=" Y(N)
1110 RETURN
```

四、例 题

1. 求正弦积分

$$\sin(x) = - \int_x^\infty \frac{\sin t}{t} dt$$

当 $x = x_0 = 0.462$ 时的值。

已知

$a_i = 0.3, 0.4, 0.5, 0.6, 0.7$

$y_i = 0.2985, 0.39646, 0.49311, 0.58813, 0.68122$

$N=4$

$X=0.462$

2. 主程序

5 DATA 4, .462

10 DATA .3, .4, .5, .6, .7

15 DATA .2985, .39646, .49311, .58813, .68122

20 GOSUB 1000

25 END

3. 计算结果

$F(.462) = .456558$

第四节 一元三点插值

一、数学方法简述

1. 已知函数 $y(x)$ 的结点值为 $x_i (i=0, 1, \dots, n)$, 其对应的函数值为 $y(x_i) = y_i (i=0, 1, \dots, n)$, 对于给定的不是结点的值 x , 选取最靠近它的三个插值点, 应用一元拉格朗日三点插值公式计算出对应的函数值 $y(x)$ 。

2. 插值公式

$$y(x) = \sum_{j=k}^{k+2} \prod_{\substack{i=k \\ i \neq j}}^{k+2} \left(\frac{x - x_i}{x_j - x_i} \right) y_i$$

式中

$$k = \begin{cases} 0 & \text{当 } x \leq x_1 \text{ 时} \\ s & \text{当 } x_s < x \leq x_{s+1}, x - x_s \geq x_{s+1} - x \text{ 时} \\ & (s=1, 2, \dots, n-2) \\ s-1 & \text{当 } x_s < x \leq x_{s+1}, x - x_s < x_{s+1} - x \text{ 时} \\ & (s=1, 2, \dots, n-2) \\ n-2 & \text{当 } x \geq x_{n-1} \text{ 时} \end{cases}$$

3. 本子程序是对 $m+1$ 个不是结点的值 $x_l (l=0, 1, \dots, m)$ 进行成组插值, 求出对应的函数值 $y_l = y(x_l)$ 。

二、子程序使用说明

1. 简单变量

N——给定的插值结点的个数减一;

M——被插值点的个数减一。

2. 数组

X(N)——存放给定的插值结点值;

Y(N)——存放给定的插值结点上的函数值;

U(M)——存放被插值点的值;

V(M)——存放插值结果。

3. 数据排列顺序

N, M, X(N), Y(N), U(M)

4. 在主程序的数据语句中, 按数据排列顺序给出具体数值后, 由 GOSUB 1000 语句转本节子程序

三、子程序

```
1000 'LAGRANGE13 INSERT VALUE
1010 READ N, M
1020 DIM X(N), V(N), U(M), V(M)
1030 FOR I=0 TO N:READ X(I):NEXT
1040 FOR I=0 TO N:READ Y(I):NEXT
1050 FOR I=0 TO M:READ U(I):NEXT
1060 FOR L=0 TO M
  { 1070 FOR K=0 TO N-3
    1080 IF U(L) <= X(K+1) THEN 1110
    1090 NEXT K
    1100 K=N-2
    1110 IF K=0 THEN 1140
    1120 IF U(L)-X(K)>=X(K+1)-U(L) THEN 1140
    1130 K=K-1
    1140 X0=X(K):X1=X(K+1):X2=X(K+2)
    1150 P=(U(L)-X1)*(U(L)-X2)/((X0-X1)*(X0-X2))
    1160 Q=(U(L)-X0)*(U(L)-X2)/((X1-X0)*(X1-X2))
    1170 R=(U(L)-X0)*(U(L)-X1)/((X2-X0)*(X2-X1))
    1180 V(L)=P*Y(K)+Q*Y(K+1)+R*Y(K+2)
    1190 LPRINT "V(" L ")=" V(L)
  } 1200 NEXT L
1210 RETURN
```

四、例 题

1. 已知函数如表 1-2 所示

表1-2

x	0.2	0.24	0.28	0.32	0.36	0.4
y	0.19867	0.2377	0.27636	0.31457	0.35227	0.38942

2. 求当 $x=0.22, 0.29, 0.38$ 时的 y 值。

3. 主程序

```
5 DATA 5, 2
```

```
10 DATA .2, .24, .28, .32, .36, .4
```

```
15 DATA .19867, .2377, .27636, .31457, .35227, .38942
```

```
20 DATA .22, .29, .38
```

```
25 GOSUB 1000
```

```
30 END
```

4. 计算结果

$V(0)=.218231$

$V(1)=.285955$

$V(2)=.370914$

第五节 二元三点插值

一、数学方法简述

1. 已知函数 $f(x, y)$ 的第一变量(x)的结点值为 x_i (不一定等距, $i=0, 1, \dots, n$), 第二变量(y)的结点值为 y_j (不一定等距, $j=0, 1, \dots, m$), 其对应结点上的函数值为 f_{ij} ($i=0, 1, \dots, n; j=0, 1, \dots, m$); 对于给定的不是结点的值(x, y), 分别选取最靠近 x 的三个点(x_q, x_{q+1}, x_{q+2})和最靠近 y 的三个点(y_p, y_{p+1}, y_{p+2}), 用二元拉格朗日插值公式计算出对应的函数值 $f(x, y)$ 。

显然, 插值结点分布越均匀越密, 其对应的函数值相差越小, 插值结果越精确。

2. 插值公式

$$f(x, y) = \sum_{i=q}^{q+2} \sum_{j=p}^{p+2} \left(\prod_{\substack{s=q \\ s \neq i}}^{q+2} \frac{x - x_s}{x_i - x_s} \right) \left(\prod_{\substack{l=p \\ l \neq j}}^{p+2} \frac{y - y_l}{y_j - y_l} \right) f_{ij}$$

式中 i, j 选取方法与一元三点插值相同。本节介绍的方法可对 $l+1$ 个不是结点的变元值 (x_k, y_k) ($k=0, 1, \dots, l$) 进行成组插值, 求出对应的函数值。

二、子程序使用说明

1. 简单变量

N —— x 方向上给定的插值结点个数减一;

M —— y 方向上给定的插值结点个数减一;

L ——被插值点的个数减一。

2. 数组

$X(N)$ ——存放 x 方向上给定的插值结点值;